// File: ERC20.sol



pragma solidity 0.5.0;

import "./IERC20.sol";
import "./SafeMath.sol";

/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * https://eips.ethereum.org/EIPS/eip-20
 * Originally based on code by FirstBlood:
 * https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 *
 * This implementation emits additional Approval events, allowing applications to reconstruct the allowance status for
 * all accounts just by listening to said events. Note that this isn't required by the specification, and other
 * compliant implementations may not do it.
 */
contract ERC20 is IERC20 {
    using SafeMath for uint256;

    mapping (address => uint256) internal _balances;

    mapping (address => mapping (address => uint256)) private _allowed;

    uint256 internal _totalSupply;

    /**
     * getter value for totalSupply
     */
    function totalSupply() public view returns (uint256) {
      return _totalSupply;
    }  

    /**
     * @dev Gets the balance of the specified address.
     * @param owner The address to query the balance of.
     * @return A uint256 representing the amount owned by the passed address.
     */
    function balanceOf(address owner) public view returns (uint256) {
        return _balances[owner];
    }

    /**
     * @dev Function to check the amount of tokens that an owner allowed to a spender.
     * @param owner address The address which owns the funds.
     * @param spender address The address which will spend the funds.
     * @return A uint256 specifying the amount of tokens still available for the spender.
     */
    function allowance(address owner, address spender) public view returns (uint256) {
        return _allowed[owner][spender];
    }

    /**
     * @dev Transfer token to a specified address
     * @param to The address to transfer to.
     * @param value The amount to be transferred.
     */
    function transfer(address to, uint256 value) public returns (bool) {
        _transfer(msg.sender, to, value);
        return true;
    }

    /**
     * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
     * Beware that changing an allowance with this method brings the risk that someone may use both the old
     * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
     * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     * @param spender The address which will spend the funds.
     * @param value The amount of tokens to be spent.
     */
    function approve(address spender, uint256 value) public returns (bool) {
        _approve(msg.sender, spender, value);
        return true;
    }

    /**
     * @dev Transfer tokens from one address to another.
     * Note that while this function emits an Approval event, this is not required as per the specification,
     * and other compliant implementations may not emit the event.
     * @param from address The address which you want to send tokens from
     * @param to address The address which you want to transfer to
     * @param value uint256 the amount of tokens to be transferred
     */
    function transferFrom(address from, address to, uint256 value) public returns (bool) {
        _transfer(from, to, value);
        _approve(from, msg.sender, _allowed[from][msg.sender].sub(value));
        return true;
    }

    /**
     * @dev Increase the amount of tokens that an owner allowed to a spender.
     * approve should be called when _allowed[msg.sender][spender] == 0. To increment
     * allowed value is better to use this function to avoid 2 calls (and wait until
     * the first transaction is mined)
     * From MonolithDAO Token.sol
     * Emits an Approval event.
     * @param spender The address which will spend the funds.
     * @param addedValue The amount of tokens to increase the allowance by.
     */
    function increaseAllowance(address spender, uint256 addedValue) public returns (bool) {
        _approve(msg.sender, spender, _allowed[msg.sender][spender].add(addedValue));
        return true;
    }

    /**
     * @dev Decrease the amount of tokens that an owner allowed to a spender.
     * approve should be called when _allowed[msg.sender][spender] == 0. To decrement
     * allowed value is better to use this function to avoid 2 calls (and wait until
     * the first transaction is mined)
     * From MonolithDAO Token.sol
     * Emits an Approval event.
     * @param spender The address which will spend the funds.
     * @param subtractedValue The amount of tokens to decrease the allowance by.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public returns (bool) {
        _approve(msg.sender, spender, _allowed[msg.sender][spender].sub(subtractedValue));
        return true;
    }

    /**
     * @dev Transfer token for a specified addresses
     * @param from The address to transfer from.
     * @param to The address to transfer to.
     * @param value The amount to be transferred.
     */
    function _transfer(address from, address to, uint256 value) internal {
        require(to != address(0));

        _balances[from] = _balances[from].sub(value);
        _balances[to] = _balances[to].add(value);
        emit Transfer(from, to, value);
    }

    /**
     * @dev Approve an address to spend another addresses' tokens.
     * @param owner The address that owns the tokens.
     * @param spender The address that will spend the tokens.
     * @param value The number of tokens that can be spent.
     */
    function _approve(address owner, address spender, uint256 value) internal {
        require(spender != address(0));
        require(owner != address(0));

        _allowed[owner][spender] = value;
        emit Approval(owner, spender, value);
    }
}

// File: ERC20Detailed.sol



pragma solidity 0.5.0;

import "./IERC20.sol";

/**
 * @title ERC20Detailed token
 * @dev The decimals are only for visualization purposes.
 * All the operations are done using the smallest and indivisible token unit,
 * just as on Ethereum all the operations are done in wei.
 */
contract ERC20Detailed is IERC20 {
    string private _name;
    string private _symbol;
    uint8 private _decimals;

    constructor (string memory name, string memory symbol, uint8 decimals) public {
        _name = name;
        _symbol = symbol;
        _decimals = decimals;
    }

    /**
     * @return the name of the token.
     */
    function name() public view returns (string memory) {
        return _name;
    }

    /**
     * @return the symbol of the token.
     */
    function symbol() public view returns (string memory) {
        return _symbol;
    }

    /**
     * @return the number of decimals of the token.
     */
    function decimals() public view returns (uint8) {
        return _decimals;
    }
}

// File: ERC20Pausable.sol



pragma solidity 0.5.0;

import "./ERC20.sol";
import "./Pauser.sol";

/**
 * @title Pausable token
 * @dev ERC20 modified with pausable transfers.
 */
contract ERC20Pausable is ERC20, Pauser {
    function transfer(address to, uint256 value) public onlyNotPaused returns (bool) {
        return super.transfer(to, value);
    }

    function transferFrom(address from, address to, uint256 value) public onlyNotPaused returns (bool) {
        return super.transferFrom(from, to, value);
    }

    function approve(address spender, uint256 value) public onlyNotPaused returns (bool) {
        return super.approve(spender, value);
    }

    function increaseAllowance(address spender, uint addedValue) public onlyNotPaused returns (bool success) {
        return super.increaseAllowance(spender, addedValue);
    }

    function decreaseAllowance(address spender, uint subtractedValue) public onlyNotPaused returns (bool success) {
        return super.decreaseAllowance(spender, subtractedValue);
    }
}

// File: IERC20.sol



pragma solidity 0.5.0;

/**
 * @title ERC20 interface
 * @dev see https://eips.ethereum.org/EIPS/eip-20
 */
interface IERC20 {
    function transfer(address to, uint256 value) external returns (bool);

    function approve(address spender, uint256 value) external returns (bool);

    function transferFrom(address from, address to, uint256 value) external returns (bool);

    function totalSupply() external view returns (uint256);

    function balanceOf(address who) external view returns (uint256);

    function allowance(address owner, address spender) external view returns (uint256);

    event Transfer(address indexed from, address indexed to, uint256 value);

    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: Ownable.sol



pragma solidity 0.5.0;


contract Ownable {

  /* Attributes */
  address private _admin;

  /* Event Definition */
  event AdminOwnershipTransferred (address indexed oldAdmin, address indexed newAdmin);


  constructor() internal {
    _admin = msg.sender;
  }

  /* GETTER MODUlE */
  function admin() public view returns(address) {
    return _admin;
  }  

  /* MODIFIER MODULE */
  /* to make function callable only invoked by admin */
  modifier onlyAdmin() {
    require (isAdmin(), "Caller is not an admin");
    _;
  }

  /* FUNCTIONAL BLOCK */  
  function isAdmin() public view returns(bool) {
    return (msg.sender == _admin);
  }

  /* relinquish control of contract by setting admin to zero address
   * after this function call onlyAdmin modifier cannot executed */
  function relinquishOwnership() public onlyAdmin {
    emit AdminOwnershipTransferred (_admin, address(0));

    _admin = address(0);      // Setting the _admin variable to point null address
  } 

  /* transfer admin privilages to another address */
  function adminTransfer(address newAdmin) public onlyAdmin {
    require (newAdmin != address(0), "Invalid address detected");

    emit AdminOwnershipTransferred (_admin, newAdmin);  

    _admin = newAdmin;
  }  
}

// File: Pauser.sol



pragma solidity 0.5.0;

import "./Ownable.sol";

contract Pauser is Ownable {
  
  /* Attributes */
  bool private _paused;

  /* Event Definition */
  event Paused (address indexed admin);
  event Unpaused (address indexed admin);


  constructor() internal {
    _paused = false;
  }

  /* GETTER MODULE */
  /* bool value to know if transfer is paused or not
   * true - contract is paused
   * false - contract not paused */
  function paused() public view returns(bool) {
    return _paused;
  }

  /* MODIFIED MODULE */
  /* to make function callable if contract is not paused */
  modifier onlyNotPaused() {
    require (!_paused, "Contract is Paused");
    _;
  }

  /* FUNCTIONAL MODULE */
  /* called by admin to pause ie. freeze all transfers */
  function pause() public onlyAdmin {
    _paused = true;

    emit Paused (msg.sender);
  }  

  /* called by admin to unpause ie. unfreeze all transfers */
  function unpause() public onlyAdmin {
    _paused = false;

    emit Paused (msg.sender);
  }
}

// File: SafeMath.sol



pragma solidity 0.5.0;

/**
 * @title SafeMath
 * @dev Unsigned math operations with safety checks that revert on error
 */
library SafeMath {
    /**
     * @dev Multiplies two unsigned integers, reverts on overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-solidity/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b);

        return c;
    }

    /**
     * @dev Integer division of two unsigned integers truncating the quotient, reverts on division by zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        // Solidity only automatically asserts when dividing by 0
        require(b > 0);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Subtracts two unsigned integers, reverts on overflow (i.e. if subtrahend is greater than minuend).
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b <= a);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Adds two unsigned integers, reverts on overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a);

        return c;
    }

    /**
     * @dev Divides two unsigned integers and returns the remainder (unsigned integer modulo),
     * reverts when dividing by zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b != 0);
        return a % b;
    }
}

// File: SlotheeToken.sol



pragma solidity 0.5.0;

import "./SafeMath.sol";
import "./ERC20Pausable.sol";
import "./ERC20Detailed.sol";

contract SlotheeToken is ERC20Pausable, ERC20Detailed {

  constructor() ERC20Detailed("SlotheeToken", "SLO", 18) public {

    // total tokens - 400 000 000 Tokens	  
    _totalSupply = 400000000000000000000000000;

    // Team - 40 000 000 Tokens
    _balances[0xEf73Cf964862231BEf1e57B2F95dA7bd1dafA4Ce] = 40000000000000000000000000;
    // Advisor and Partners - 40 000 000 Tokens
    _balances[0xacAade0E303fE98Bd73bD75daD312d9Bc7bfd695] = 40000000000000000000000000;
    // Bounty and Airdrop - 20 000 000 Tokens
    _balances[0x76439EE8Ec9c5908dbCAA7cdBB4Ef715C9d8b710] = 20000000000000000000000000;
    // Reward - 20 000 000 Tokens
    _balances[0xDD90b2f934D843684647f5e7cDc30b298D3287fF] = 20000000000000000000000000;
    // Holding address - 280 000 000 Tokens
    _balances[0x1B8A6c794ec32fe9DC47bD124ec9c6C83B98c0c9] = 280000000000000000000000000;
    
  }  
}